

# Logic and Mechanized Reasoning

## Using SMT Solvers

**Marijn J.H. Heule**

**Carnegie  
Mellon  
University**

SMT-LIB

Example: Magic Squares

Application: Verification

# SMT-LIB

Example: Magic Squares

Application: Verification

## SMT-LIB: Introduction

Consists of five blocks:

- ▶ theory (`set-logic ...`), e.g. `QF_UF` and `QF_LIA`
- ▶ variables, functions, and types (`declare-const ...`)
- ▶ a list of constraints (`assert ...`)
- ▶ solving the problem (`check-sat`)
- ▶ termination the solver (`exit`)

## SMT-LIB: Introduction

Consists of five blocks:

- ▶ `theory (set-logic ...)`, e.g. `QF_UF` and `QF_LIA`
- ▶ variables, functions, and types (`declare-const ...`)
- ▶ a list of constraints (`assert ...`)
- ▶ solving the problem (`check-sat`)
- ▶ termination the solver (`exit`)

Variable and functions:

- ▶ `(declare-const name type)`
- ▶ `(declare-fun name (inputTypes) outputType)`
- ▶ `(define-fun name (inputTypes) outputType (body))`

## SMT-LIB: QF\_UF example

### Example

Does there exist a satisfying assignment for  $p \wedge \neg p$ ?

```
(set-logic QF_UF)
(declare-const p Bool)
(assert (and p (not p)))
(check-sat) ; should be UNSAT
(exit)
```

## SMT-LIB: QF\_LIA example

### Example

Does there exist an integer  $x$  that is larger than an integer  $y$ ?

```
(set-logic QF_LIA)
(declare-const x Int)
(declare-const y Int)
(assert (> x y))
(check-sat) ; should be SAT
(get-model)
(exit)
```

SMT-LIB

Example: Magic Squares

Application: Verification



## Magic Squares: Introduction

A  $n \times n$  square is called a magic square if each number from 1 to  $n^2$  occurs uniquely and the sum of all rows, columns, and diagonals is the same:  $(n^3 + n)/2$

2	7	6	→	15
9	5	1	→	15
4	3	8	→	15

15 ↙    ↓    ↓    ↓    ↘ 15

15    15    15    15    15

## Magic Squares: Linear Arithmetic

```
(set-logic QF_LIA)
(declare-const m_0_0 Int)
(declare-const m_0_1 Int)
...
(declare-const m_2_2 Int)
(assert (and (> m_0_0 0) (<= m_0_0 9)))
(assert (and (> m_0_1 0) (<= m_0_1 9)))
...
(assert (and (> m_2_2 0) (<= m_2_2 9)))
(assert (distinct m_0_0 m_0_1 m_0_2 m_1_0
                  m_1_1 m_1_2 m_2_0 m_2_1 m_2_2))
(assert (= 15 (+ m_0_0 m_0_1 m_0_2)))
(assert (= 15 (+ m_1_0 m_1_1 m_1_2)))
...
(assert (= 15 (+ m_2_0 m_1_1 m_0_2)))
(check-sat)
(get-model)
(exit)
```

## Magic Squares: Bitvectors

```
(set-logic QF_BV)
(declare-const m_0_0 (_ BitVec 16))
(declare-const m_0_1 (_ BitVec 16))
...
(declare-const m_2_2 (_ BitVec 16))
(assert (and (bvugt m_0_0 #x0000) (bvule m_0_0 #x0009)))
(assert (and (bvugt m_0_1 #x0000) (bvule m_0_1 #x0009)))
...
(assert (and (bvugt m_2_2 #x0000) (bvule m_2_2 #x0009)))
(assert (distinct m_0_0 m_0_1 m_0_2 m_1_0
                  m_1_1 m_1_2 m_2_0 m_2_1 m_2_2))
(assert (= #x000f (bvadd m_0_0 m_0_1 m_0_2)))
(assert (= #x000f (bvadd m_1_0 m_1_1 m_1_2)))
...
(assert (= #x000f (bvadd m_2_0 m_1_1 m_0_2)))
(check-sat)
(get-model)
(exit)
```

## Magic Squares: Theory Differences

The SMT-LIB formula for QF\_BV and QF\_LIA looks similar...

## Magic Squares: Theory Differences

The SMT-LIB formula for QF\_BV and QF\_LIA looks similar...

The QF\_LIA abstracts the problem by turning  $(m_{2,2} > 0)$  into a literal  $(p \leftrightarrow m_{2,2} > 0)$

## Magic Squares: Theory Differences

The SMT-LIB formula for QF\_BV and QF\_LIA looks similar...

The QF\_LIA abstracts the problem by turning  $(\sum_{i,j} m_{i,j} > 0)$  into a literal  $(p \leftrightarrow m_{2,2} > 0)$

QF\_LIA: the solver applies (exponentially) **many** SAT calls

## Magic Squares: Theory Differences

The SMT-LIB formula for QF\_BV and QF\_LIA looks similar...

The QF\_LIA abstracts the problem by turning  $(> m_{2,2} 0)$  into a literal  $(p \leftrightarrow m_{2,2} > 0)$

QF\_LIA: the solver applies (exponentially) **many** SAT calls

When using QF\_BV, the solver applies **bitblasting**: every bit in each bitvector is turned into a propositional variable. Each constraint, such as  $(> m_{2,2} 0)$  is turned into many clauses.

## Magic Squares: Theory Differences

The SMT-LIB formula for QF\_BV and QF\_LIA looks similar...

The QF\_LIA abstracts the problem by turning  $(> m_{2,2} 0)$  into a literal  $(p \leftrightarrow m_{2,2} > 0)$

QF\_LIA: the solver applies (exponentially) **many** SAT calls

When using QF\_BV, the solver applies **bitblasting**: every bit in each bitvector is turned into a propositional variable. Each constraint, such as  $(> m_{2,2} 0)$  is turned into many clauses.

QF\_BV: the solver applies a **single** SAT call



## Magic Squares: Theory Differences

The SMT-LIB formula for QF\_BV and QF\_LIA looks similar...

The QF\_LIA abstracts the problem by turning  $(\> m_{2,2} 0)$  into a literal  $(p \leftrightarrow m_{2,2} > 0)$

QF\_LIA: the solver applies (exponentially) **many** SAT calls

When using QF\_BV, the solver applies **bitblasting**: every bit in each bitvector is turned into a propositional variable. Each constraint, such as  $(\> m_{2,2} 0)$  is turned into many clauses.

QF\_BV: the solver applies a **single** SAT call

**Compare:**  $n \geq 5$  is hard for QF\_LIA,  $n \leq 10$  is easy for QF\_BV

## Magic Squares: Demo

SAT with assignment:

$m_{2_2} \mapsto 2$

$m_{2_1} \mapsto 9$

$m_{2_0} \mapsto 4$

$m_{1_2} \mapsto 7$

$m_{1_1} \mapsto 5$

$m_{1_0} \mapsto 3$

$m_{0_2} \mapsto 6$

$m_{0_1} \mapsto 1$

$m_{0_0} \mapsto 8$

Square:

8 1 6

3 5 7

4 9 2

SMT-LIB

Example: Magic Squares

Application: Verification

# Verification: Equivalence Checking

SAT and SMT solvers are crucial for verification tasks

- ▶ Equivalence checking
- ▶ Bounded model checking


Equivalence checking:

- ▶ Are two hardware/software designs functionally equivalent?
- ▶ Does any input to both produces the same output?
- ▶ Typically one is unoptimized and the other is optimized

## Verification: Popcount

Popcount: count the number of 1's in a bitvector

```
int popCount32 (unsigned int x) {  
    x = x - ((x >> 1) & 0x55555555);  
    x = (x & 0x33333333) + ((x >> 2) & 0x33333333);  
    x = ((x + (x >> 4) & 0xf0f0f0f) * 0x1010101) >> 24;  
    return x; }  
}
```



## Verification: General Setup

```
(set-logic QF_BV)
(declare-const x (_ BitVec 32))

(define-fun fast ((x (_ BitVec 32))) (_ BitVec 32)
  ...

(define-fun slow ((x (_ BitVec 32))) (_ BitVec 32)
  ...

(assert (not (= (fast x) (slow x))))
(check-sat) ; expect UNSAT
(exit)
```

## Verification: Specification

```
(define-fun slow ((x (_ BitVec 32))) (_ BitVec 32)
  (bvadd
    (ite (= #b1 ((_ extract 0 0) x)) #x00000001 #x00000000)
    (ite (= #b1 ((_ extract 1 1) x)) #x00000001 #x00000000)
    (ite (= #b1 ((_ extract 2 2) x)) #x00000001 #x00000000)
    . . .
    (ite (= #b1 ((_ extract 30 30) x)) #x00000001 #x00000000)
    (ite (= #b1 ((_ extract 31 31) x)) #x00000001 #x00000000)))
```

## Verification: Code conversion

```
int popCount32 (unsigned int x) {  
    x = x - ((x >> 1) & 0x55555555);  
    x = (x & 0x33333333) + ((x >> 2) & 0x33333333);  
    x = ((x + (x >> 4) & 0xf0f0f0f) * 0x1010101) >> 24;  
    return x; }
```

```
(define-fun line1 ((x (_ BitVec 32))) (_ BitVec 32)  
  (bvsub x (bvand (bvlshr x #x00000001) #x55555555)))
```

```
(define-fun line2 ((x (_ BitVec 32))) (_ BitVec 32)  
  (bvadd (bvand x #x33333333)  
    (bvand (bvlshr x #x00000002) #x33333333)))
```

```
(define-fun line3 ((x (_ BitVec 32))) (_ BitVec 32)  
  (bvlshr (bvmul (bvand (bvadd (bvlshr x #x00000004)  
    x) #x0f0f0f0f) #x01010101) #x00000018)))
```

```
(define-fun fast ((x (_ BitVec 32))) (_ BitVec 32)  
  (line3 (line2 (line1 x))))
```



## Verification: Demo

```
#eval (do
  let out ← callZ3 popcount (verbose := true)
  : IO Unit)
```

Solver replied:  
unsat